Table of 2D and 3D Fourier transforms There is good book "Tables of Integral Transforms" by Bateman & Erdélyi where a lot of commonly used Fourier integrals are collected.
Is there a similar comprehensive collection of 2D and 3D Fourier transforms of functions occurring in physics and mathematics?
Update:
Below I give a few examples of Fourier transforms, which (and similar to which) I was hoping to find.
$$
\frac1{\omega^2-k_x^2-b^2},\;\;\;\;
\frac1{\omega^2-k_x^2-k_y^2-b^2}, \; b\in\mathbb{C}
$$
$$
\exp\left\{{\rm i} z \sqrt{\omega^2 - k_x^2}\right\}, \;\;\;\;
\exp\left\{{\rm i} z \sqrt{\omega^2 - k_x^2 - k_y^2}\right\}
$$
The Fourier transform here is to be calculated with respect to $\omega$, $k_x$ and $k_y$.
All these integrals can be calculated and have quite simple closed form.
 A: For 2D and 3D transforms of functions with radial symmetry, the Fourier transform reduces to a Hankel Transform which is one dimensional in the radial variable.  See the first part of this answer https://math.stackexchange.com/a/3029986/441161 and https://en.wikipedia.org/wiki/Hankel_transform .  Chapter VIII of the book you reference has tables of Hankel Transforms.
For 2D and 3D transforms with no special symmetry, the Fourier transform is taken by treating each cartesian coordinate variable separately.  So again a table of 1D Fourier transforms will usually suffice.
In Bracewell's book The Fourier Transform and Its Applications, Bracewell does have a chapter where he discusses the multi-dimensional Fourier Transform, its relation to the Hankel Transform, and a small selection of 2D Fourier Transform examples depicted graphically.
A: Based on Andy Walls' answer I threw together some Mathematica code to implement the Fourier transform of a radial function as a Hankel transform.
f[r_] := 1/r
n = 2;
(2 \[Pi])^(n/2)/k^(n/2 - 1)
  HankelTransform[f[r] r^(n/2 - 1), r, k, n/2 - 1] 

Here I used the conventions from the wikipage on Hankel transforms for which the Fourier transform reads
$$F(\mathbf k)=\int_{\mathbb R^n}\mathrm d\mathbf rf(\mathbf r)e^{-i\mathbf k\cdot \mathbf r}$$
This means the code above is equivalent to
FourierTransform[1/Sqrt[x^2 + y^2], {x, y}, {kx, ky}, 
 FourierParameters -> {1, -1}]

Note that for $n=3$ the regular Fourier transform is so slow it doesn't seem to evaluate but the Hankel transform evaluates instantaneously. Also I checked if everything works but if you catch any errors please let me know.

Update: I modified the function so it works with a more general Fourier transform (that handles all the conventions) 
$$\hat f(\mathbf k)=\left(\frac{|b|}{(2\pi)^{1-a}}\right)^{n/2}\int \text{d}^n\mathbf x\, e^{ib\mathbf k\cdot\mathbf x}f(x)$$
in accordance with [the Mathematica definition of Fourier parameters][2]
The updated code is now
Options[RadialFourierTransform] = {FourierParameters -> {1, -1}}
RadialFourierTransform[expr_, r_, k_, n_Integer, 
  opts : OptionsPattern[]] := 
 With[{a = OptionValue[FourierParameters][[1]], 
   b = OptionValue[FourierParameters][[2]]},
  (2 \[Pi])^(n/2)/q^(n/2 - 1) (Abs[b]/(2 \[Pi])^(1 - a))^(n/2)
       HankelTransform[expr r^(n/2 - 1), r, q, n/2 - 1] /. q -> b k //
     Normal // Simplify
  ]

How to use:
RadialFourierTransform[1/(\[Omega]^2 - r^2), r, k, 3, 
  FourierParameters -> {1, -1}] // FullSimplify

Output:
$$\frac{2\pi^2\cos k\omega}{k}$$
